

How to pick shares in five easy steps

Without prior knowledge, relying on elementary charting and stats

I am assuming that you have limited knowledge of analytics and that you are trying to buy some shares to protect the value of your money, such as opening an ISA shares account. Your plan is not to sell these shares too often, although you might be forced to sell some of them at one point for a variety of reasons. We will ignore this part and leave it for some other tutorial. In this tutorial, we will only focus on the buying side.

So, you want to buy some shares, but how do you decide which shares to buy and when? This is a real art and goes beyond science and analytics. What we are going to show here is a very simple approach. It does not require you to understand P/E, EPS, debt ratios (though it would help!), or any of the analytics. It just requires access to data, some basic manipulations and charting. That's it. Let's go step by step.

Where do you get the data, how do you analyse the data and how do you visualise your data? The answer to all these questions is Microsoft Excel. In **Appendix A1** we provide the details as to how to download the real-time shares data from stock exchanges.

For the purposes of this tutorial, we just picked Oracle data from the New York Stock Exchange and downloaded the values for the last 2,000 daily values, which is some 5 years' worth of data, starting from 08/02/2019 until 30/07/2024.

Now we have data in our spreadsheet, we can start some of the basic analytics and charting. The first step is to produce a line graph of all the closing values over the last 5 years and fit a trend (see Fig. 1). For simplicity reasons, we will be using a linear trend, though this is not always the best option. In **Appendix A2** you have the details as to how to do this.




Fig. 1

These are the closing price values over the last 5 years. Looking at the movements of this stock in Fig. 1, it is obvious to see that it is an upward-moving stock.

The long-term graph (5 years) and the fitted trend provide a good indication of the general movements of the stock and the dynamics of this stock. Although we can see a consistent long-term upward trend, we also see some short-term or even medium-term variations. Sometimes the difference between the local peak and local trough is \$20, \$30 or even \$40. This implies that it is very important to know when to buy these shares.

By just looking at the chart in Fig. 1, would you decide to buy these shares? Probably yes. It is a reputable company that continues to deliver growth. Some variability hints that we need to be careful to decide at what point to buy the shares, but so far, on a binary scale of Go / No-go, this is a definite Go. The growth pattern indicates that in the long run, you are likely to protect your money from erosion.

Regarding the point as to when to buy these shares, it will take a few more steps to have a more considered answer. We'll gradually tackle this question. The first point to consider is if this is a good time to buy it at all. From Fig. 1 you can see that there were times when the shares took a downward trend. The question is, if you buy them now, are they going to continue to grow? The graph below in Fig. 2 provides some clues.




Fig. 2

The chart above in Fig. 2 indicates that we have now switched from a macro level (5 years view) to a more tactical level, and I picked here only the last 90 days of the movements of the same share.

We are showing the closing values of this stock, together with moving averages for 50 trading days (50MA in red) and moving averages for 200 trading days (200MA in green). A general rule is that if the 50MA line is above 200MA, then it indicates that the shares are in the growth phase. Depending on where they are in this phase, it would make sense to buy them. Otherwise, it would make sense to sell them, again depending on where they are in this cycle.

Our chart shows that over the whole 90-day period, the 50MA line was above the 200MA line, so it is in principle a good time to buy. When the short-term moving averages (50MA) are higher in value than the more long-term moving averages (200MA) this indicates that more recent history contains higher values than the more distant history. In a word, the share prices are currently in a growth mode.

How the moving averages are calculated is given in **Appendix A3**.

OK, now we know that this whole 90-day period was a good period to buy, but at what price, or at what point in time? We are trying to learn something that could be applied when making future decisions to buy.

Still looking at the previous chart in Fig. 2, we can see that 90 days ago on 21/03/2024 the closing price was \$129.01 and today on 30/07/2024 it is \$135.67. In between, there was a lot of ups and downs. Let's produce a chart that will show us by what percentage the price changed every day when compared to a fixed price of 90 days ago. The graph below in Fig. 3 shows this.




Fig. 3

These simple calculations are explained in **Appendix A4**.

The chart shows that today's price (on 30/07/2024) is only 5.16% higher than the price 90 days ago. However, at one point the difference was at the maximum percentage point of 12.42%. At the same time, at another point, the price shows a minimum drop of -11.83%, when compared to the origin point. What does this tell us? If we could have figured out the date when it was -11.83%, this would have been a good day to buy. Equally, if we knew that 12.42% was going to be the maximum growth, we could have sold the shares and made a nice profit. Let's see if we can figure this one out in real time to make a smart decision today.

For this purpose, we will use something called the Relative Strength Index (RSI). Details of how the calculations are done are given in **Appendix A5**, and we are going to explore how to benefit from this method.

The chart below in Fig. 4 shows RSI for the last 90 days of our data. When the RSI line crosses the horizontal line marking 70, this is the time to sell as the prices will start to fall. Equally, when the line drops below 30, this is the time to buy as the prices will start to grow.

We can see that between 21-24/05/2024 we had warnings that the price would start dropping. The second interval of warnings that the price would go down was between 21/06/2024 and 03/07/2024. These two points in time would have been the right moments to sell. However, as we are looking to buy the shares, we are interested in RSI below 30.

We only had one clear warning at the beginning, i.e. around 21/03/2024 and one weak warning on 19/07/2024. Due to such a weak second signal, and due to the fact that at the moment the changes in prices are stable, the best decision is to wait before we decide to buy this stock.




Fig. 4

The above chart should always be looked at in the context of the actual share movements. To illustrate this, we overlaid the RSI chart with the closing price values line chart, as in Fig. 5.




Fig. 5

To make it work, we had to introduce the secondary axis for the closing prices line. Let's look at four areas that are outside the boundaries.

We put the green circle on the RSI chart indicating that for the first time in our range, the closing values will fall. On the closing prices section (dotted line), to demonstrate that this actually happened, we inserted a green line where the prices started to fall.

The same thing with the orange circle. It also indicated another fall, and looking at the orange superimposed dotted line, this happened too. As we are interested in buying, let's see what happened when the RSI dropped below the 30 line.

At the very beginning of our chart, we already had an indication that prices would start to rise. We marked this on the RSI with a purple circle, followed by a purple ascending line on the dotted closing price line. At the very end of the chart we have another breach below the 30 line, and we also circled it with a purple circle. Unfortunately, this is the end of our range, we do not know if the prices will start to rise again. The signal is weak, i.e. it just touched the 30 RSI line, so it might be a false alarm.

What do we conclude? If this is a genuine alarm, we should definitely buy it as the prices will potentially go up. However, as the signal is weak, perhaps it is better to hold back and either wait a few days or do some other analysis.

RSI chart served the purpose of giving us a clue that the prices might go up or down. However, is there a method that can help us identify the actual turning point? The answer is, yes, and this can be achieved by identifying a turning point. You cannot predict the future turning points, but you can figure out if the new turning point has just begun.

A quick definition, a turning point is the price level at which the prices change direction, i.e. start going up or down. In other words, the local trend changes. Let's see again our last 90 days of closing prices together with the turning points mapped. Details on how to calculate turning points are given in **Appendix A6**.

This chart in Fig. 6 shows us the points in time when significant up or down intervals started. For example, on 25/03/2024 the stock value closed at \$126.08, which indicated that a downward trend had started. If you intended to buy these shares, you should have watched these movements from that date and probably bought them some time after 26/04/2024 when the trend changed again.




Fig. 6

Today, i.e. at the end of this interval, the price is \$135.67. If you bought them on 30/04/2024 for \$113.75, you would have made a profit of 19.27% in less than 3 months.

Same logic for selling and the upward trend. On 09/07/2024 when the price was \$140.68, it started to go down. If you intended to sell, this would have been the right moment in time to sell.

A turning point method gives you a clue as to when the local trend changes, but it does not necessarily pinpoint exactly the day when you should buy or sell. It is based on contrasting the current price with the moving average price of five days prior to today's price. For this reason, it does not indicate precisely the day when the turning point occurs. It has a delay of a day or three, but it is a good indicator that the trend has changed. For this reason, you still need to make a judgement call as to when is the day you are going to pull the trigger and buy or sell.

As we are trying to define the best point to buy, given that we have the picture in Fig. 6, the question is, when do we buy this stock? The current date is 30/07/2024 and we still do not see a change to an upward trend, so I would wait a few days to see this signal changing on our chart and then make a move to buy this stock. This brings our five-step analysis of whether we should buy these shares to a conclusion.

In summary, we asked ourselves five key questions and provided a simple methodology to answer these questions. The questions were:

Question 1: Shall I buy this stock?

Answer 1: Look at the long-term trend (5-10 years) and see if it follows an upward trend. Or, it follows a horizontal line, but has up-and-down oscillations that you might exploit.

Question 2: Are these shares currently in a growth mode?

Answer 2: Look at the last 90 days and compare the closing prices with 50MA and 200MA lines. If 50MA is above 200MA, then generally it is a good time to buy as the shares are following a mid-term upward trend.

Question 3: How stable or subject to variability are these shares?

Answer 3: Look at the daily percentage changes from the fixed point of 90 days ago. This will tell you how much variability have these prices been exposed to over the last 3 months. If the variability is small, you can buy any time. If variability is significant, you should pause and decide on the price level to buy.

Question 4: Is the local price trend likely to change?

Answer 4: Calculate RSI and see if the price is overvalued (it will start to drop shortly) or undervalued (it will start to increase shortly).

Question 5: At what point in time, or at what price level, should I buy?

Answer 5: Create a turning point chart and see if the prices are currently in a downward trend interval, or an upward trend interval. If it is an upward trend interval, wait for the prices to change the direction and hit the local rock bottom. If it is a downward trend interval, make a judgement call as to when to buy on the cusp, or just after, the new turning point.

Branko Pecar

Summer 2024

Appendix

A1 How to download the stocks and shares data

If you have a Microsoft 365 subscription, then your Excel has a Stocks function in the Data tab. See Fig. A1.1




Fig. A1.1

If you type in cell A1 (see Fig. A1.2), for example, the word Oracle, and then click on the Stocks symbol shown above in Fig. A1.1, you will get some options that will appear on the right side of your screen (see the right of Fig. A1.2).




Fig. A1.2

You need to select which “Oracle” data you want. We clicked on the first one that will pull the data from Oracle Corporation quoted on the New York Stock Exchange. As soon as you click, the value in cell A1 will change its appearance, indicating that Excel is ready to get the data for this specific shares (see the bottom left of Fig. A1.2)

In cell G1 we used function =A1.Price, which means, “get me the latest price for the stocks listed in cell A1”, which is Oracle. We instantly get the value of \$139.45 (Fig. A1.4, cell G1). Note that the value of this cell will change every time the price changes on the stock exchange. If the Exchange is closed, you get the last known Close price value.




Fig. A1.3

Note that while the cursor was in cell G1 in Fig. A1.3, as soon as we typed =A1. (in other words, put a dot after the number 1), Excel immediately gave us a list of indicators we could get from the stock exchange. We picked Price, but we could have picked any other option.

Also, in cell A5 (see Fig. A1.4) we entered the function =STOCKHISTORY(A1,TODAY()-10,TODAY(), 0,1,0,2,3,4,1,5). In other words, get the last 10 days of Oracle stocks (listed in A1) and show me the Date, Open, High, Low, Close and Volume data. In Fig. A1.4 you can see the result.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	🏠 ORACLE CORPORATION (XNYS:ORCL)						\$ 139.45	=A1.Price					
2													
3													
4													
5	Date	Open	High	Low	Close	Volume		=STOCKHISTORY(A1,TODAY()-10,TODAY(),0,1,0,2,3,4,1,5)					
6	22/07/2024	\$ 139.74	\$ 140.68	\$ 139.12	\$ 140.17	5,387,176							
7	23/07/2024	\$ 141.30	\$ 143.22	\$ 141.21	\$ 143.11	6,697,410							
8	24/07/2024	\$ 141.70	\$ 142.66	\$ 138.40	\$ 138.77	8,018,807							
9	25/07/2024	\$ 140.15	\$ 141.07	\$ 137.70	\$ 137.82	7,654,625							
10	26/07/2024	\$ 138.87	\$ 139.96	\$ 138.38	\$ 139.01	5,139,377							
11	29/07/2024	\$ 139.27	\$ 139.78	\$ 137.83	\$ 138.31	4,054,647							
12	30/07/2024	\$ 139.07	\$ 139.57	\$ 134.96	\$ 135.67	6,505,701							
13	31/07/2024	\$ 139.69	\$ 141.55	\$ 138.66	\$ 139.45	7,188,043							

Fig. A1.4

As you can see from the main text, instead of just 10 days you can say 2,000 days (5 years or so), which is exactly what we did. Note that 2,000 days is equivalent to some 1,377 trading days, as stock exchanges do not operate on Saturdays, Sundays and public holidays.

After you save a spreadsheet with these functions, every time you open it again, it will fetch the latest data, per your functions. If you keep it open during the trading day, you will notice that it updates itself every 15 minutes, unless you pay a special subscription to get the data at a higher frequency.

A2 How to fit the trend to your data

I assume you know how to create a line graph from your data. The easiest way to add the trendline to a line chart is to use the graphical method. Below is the chart in Fig. A2.1 containing the line graph of the closing price values over the whole downloaded period.




Fig. A2.1

You just need to right-click by touching the line graph and the option box opens, as in Fig. A2.2.




Fig. A2.2

Now, select "Add Trendline..."

This opens another option box on the right-hand side of your screen (see Fig. A2.3). Select “Linear” and select “Display Equation on chart”.




Fig. A2.3

Your graph will instantly change and it will look as in Fig. A2.4. That’s it!




Fig. A2.4

The linear trend equation is $y=41.681+0.0551x$. Remember, “y” represents your trend values and “x” are simple sequential numbers representing the days. In other words, you can create another column and call it “x”. The beginning of this column coincides with the beginning of your data set.

The graph below in Fig. A2.5 shows just the first 5 rows and the last 7 rows of the whole range. As you can see, in column B we inserted sequential numbers 1, 2, 3, ..., 1375, 1376 and 1377. They just represent the number of days.

	A	B	C	D	E	F	G
1						a=	41.681
2	Date	x	Close	Trend		b=	0.0551
3	08/02/2019	1	\$ 51.03	41.7361	=G\$1+B3*G\$2		
4	11/02/2019	2	\$ 51.23	41.7912			
5	12/02/2019	3	\$ 51.22	41.8463			
6	13/02/2019	4	\$ 51.42	41.9014			
7	14/02/2019	5	\$ 51.48	41.9565			
1373	22/07/2024	1371	\$ 140.17	117.2231			
1374	23/07/2024	1372	\$ 143.11	117.2782			
1375	24/07/2024	1373	\$ 138.77	117.3333			
1376	25/07/2024	1374	\$ 137.82	117.3884			
1377	26/07/2024	1375	\$ 139.01	117.4435			
1378	29/07/2024	1376	\$ 138.31	117.4986			
1379	30/07/2024	1377	\$ 135.67	117.5537	=G\$1+B1379*G\$2		

Fig. A2.5

From the equation, $y=41.681+0.0551x$, the number 41.681 is the so-called intercept. This is the initial value from which the trend will start before the first period, i.e. for $x=0$. The number 0.0551 is the slope. This is the amount (in this case \$) by which this trend will increment for every step it takes. Another way to say this is to conclude that on average, the price is every day \$0.0551 higher than it was the day before.

We put the value of the intercept in cell G1 and the value of the slope in cell G2 in Fig. A2.5. This enabled us to calculate in column D the trend values for every observation. We could have used a more elegant version, such as Excel’s =TREND() function, but this approach makes things easier to understand.

For $x=1$, the trend value is 41.7361 (see cell D3 in Fig. A2.5), and for the last data point, i.e. for $x=1377$, the trend value is 117.5537 (cell D1379 in Fig. A2.5).

As we can see, the trend does not perfectly match our actual numbers. But this is not our objective here. We just want to get a feel for the general direction and how steep these movements are.

From the graph, we can see that the trend is positive, i.e. upwards and we can see how steep this growth has been over the last five years. That’s all we need here.

A3 How to calculate 50-day (50MA) and 200-day (200MA) moving averages

The moving average (MA) is a rolling average for a certain number of data in a given period. If we want 50MA, we can in the last cell of the spreadsheet, in our case cell C1379 in Fig. A3.1, insert a formula =AVERAGE(B1330:B1379). For 200MA, we insert in cell D1379 a formula =AVERAGE(B1180:B1379). All we have to do is copy these formulae upwards.

	A	B	C	D	E	F	G
1	🏠 ORACLE CORPORATION (XNYS:ORCL)						
2	Date	Close	50 MA	200MA			
1373	22/07/2024	\$ 140.17	132.42	118.01			
1374	23/07/2024	\$ 143.11	132.95	118.19			
1375	24/07/2024	\$ 138.77	133.39	118.34			
1376	25/07/2024	\$ 137.82	133.82	118.48			
1377	26/07/2024	\$ 139.01	134.18	118.63			
1378	29/07/2024	\$ 138.31	134.51	118.77			
1379	30/07/2024	\$ 135.67	134.78	118.90	=AVERAGE(B1180:B1379)		
1380			=AVERAGE(B1330:B1379)				

Fig. A3.1

Do not forget, the last 200MA cell you can calculate is for cell D201, as this is the last 200 interval of values we can capture. For 50MA, the last cell is C52, for the same reason.

A4 How to calculate percentage from the fixed base 90 days prior to the last data point

In Fig. A4.1 we are only showing the first five and the last five rows of the 90-day interval. Column C contains daily changes, which when charted as a line graph take the shape of white noise. See chart Fig. A4.2 on the left. This chart is not helpful.

However, if we compare daily changes with some fixed base, say the price 90 days ago, and then calculate the cumulative values, this is much more revealing and useful. We have done this in columns D and E in Fig. A4.1.

This calculation will tell us what is the percentage difference between any daily value in this 90-day interval and the current closing value.

To achieve this, we first assign zero in cell D1290 (see Fig. A4.1). From this cell down (cell D1291), we create a formula =100*(B1291-B1290)/\$B\$1290, which is copied down to cell D1379.

In column E, cell E1291 contains cumulative values, and the last cell in this column, i.e. E1379, has formula =D1379+E1378.

From the chart on the right of Fig. A4.2, we can see the percentages quantifying by how much is every current closing price different from the starting price at the beginning of the 90-day interval. You can read this chart as: If I bought these shares on 21/03/2024 at \$129.09, how much theoretical profit would I get on any of the last 90 days?

	A	B	C	D	E	F
1	🏠 ORACLE CORPORATION (XNYS:ORCL)					
2	Date	Close	Daily chng	Fixed chng	Cum Growth	
1289	20/03/2024	\$ 129.24				
1290	21/03/2024	\$ 129.01	0	0	0	
1291	22/03/2024	\$ 127.79	-0.95	-0.95	-0.95	
1292	25/03/2024	\$ 126.08	-1.34	-1.33	-2.27	
1293	26/03/2024	\$ 126.47	0.31	0.30	-1.97	
1294	27/03/2024	\$ 125.27	-0.95	-0.93	-2.90	
1375	24/07/2024	\$ 138.77	-3.03	-3.36	7.57	
1376	25/07/2024	\$ 137.82	-0.68	-0.74	6.83	
1377	26/07/2024	\$ 139.01	0.86	0.92	7.75	
1378	29/07/2024	\$ 138.31	-0.50	-0.54	7.21	
1379	30/07/2024	\$ 135.67	-1.91	-2.05	5.16	
1380			$=100*(B1379-B1378)/B1378$			
1381			$=100*(B1379-B1378)/\$B1290			
1382			$=D1379+E1378$			
1383						
1384						
1385					12.42 =Max	
1386					-11.83 =Min	

Fig. A4.1

Column E shows the rolling price differences between two consecutive days, compared to the first value of the 90-day interval, i.e. the value on 21/03/2024.




Fig. A4.2

So, the graph on the right-hand side of Fig. A4.2 shows you how much is every day during this interval a closing price different when compared to the initial price at the beginning of this 90-day interval.

Although the current price is only 5.16% higher than the initial one, we can see that at one point the price was 12.42% higher. Equally, at another point, the price was 11.83% below the initial one. This implies that if we could identify similar points in time in the future, we can decide to either sell or buy stocks at the price which is more advantageous for us.

A5 How to calculate RSI and create RSI charts

Relative Strength Index (RSI) has the formula:

$$RSI = 100 - \frac{100}{1 + \left(\frac{\text{Average 14 days gains}}{\text{Average 14 days losses}} \right)}$$

The value of RSI goes between 0 and 100, with the equilibrium around 50. If the index is above 50, then the stock is increasing. Furthermore, if RSI goes above 70, the stock is considered to be overbought and the upward price trend is about to reverse in a downward direction. If you do not sell, you will potentially lose money.

If the RSI starts dropping below 50, the stock will lose momentum and the price will fall. If it is below 30, the stock is considered to be oversold, or overvalued, and the downward price trend is about to change and start going upwards. If you buy, you will potentially make a profit.

In a way, the level of RSI of 30 and 70 is similar to the support and resistance levels that some other indicators and stock charts use.

In Fig. A5.1 we have the details as to how to calculate RSI. We are showing formulae for just the last row, 1379, but the same formulae apply to all the last 90 rows.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1	ORACLE CORPORATION (XNYS:ORCL)													
2	Date	Close	Gains	Losses	Avg Gain	Avg Loss	RSI	Interpret	Low 30	High 70				
1359	01/07/2024	\$ 143.09	1.89		4.645	1.189	79.621	The price will go DOWN	30	70				
1360	02/07/2024	\$ 143.28	0.19		4.009	1.160	77.557	The price will go DOWN	30	70				
1361	03/07/2024	\$ 144.38	1.1		3.645	1.237	74.660	The price will go DOWN	30	70				
1362	05/07/2024	\$ 144.83	0.45		1.639	1.237	56.982	Stable	30	70				
1363	08/07/2024	\$ 145.03	0.2		1.479	1.355	52.186	Stable	30	70				
1364	09/07/2024	\$ 140.68		4.35	1.479	1.793	45.195	Stable	30	70				
1365	10/07/2024	\$ 142.07	1.39		1.280	1.793	41.649	Stable	30	70				
1366	11/07/2024	\$ 142.76	0.69		0.987	1.793	35.492	Stable	30	70				
1367	12/07/2024	\$ 144.77	2.01		1.089	1.806	37.617	Stable	30	70				
1368	15/07/2024	\$ 143.07		1.7	1.089	1.864	36.878	Stable	30	70				
1369	16/07/2024	\$ 142.61		0.46	1.089	1.634	39.993	Stable	30	70				
1370	17/07/2024	\$ 139.26		3.35	1.089	2.160	33.518	Stable	30	70				
1371	18/07/2024	\$ 138.03		1.23	1.089	2.218	32.930	Stable	30	70				
1372	19/07/2024	\$ 138.56	0.53		0.947	2.218	29.921	The price will go UP	30	70				
1373	22/07/2024	\$ 140.17	1.61		1.006	2.218	31.203	Stable	30	70				
1374	23/07/2024	\$ 143.11	2.94		1.111	2.218	33.373	Stable	30	70				
1375	24/07/2024	\$ 138.77		4.34	1.213	2.572	32.056	Stable	30	70				
1376	25/07/2024	\$ 137.82		0.95	1.228	2.340	34.408	Stable	30	70				
1377	26/07/2024	\$ 139.01	1.19		1.320	2.340	36.066	Stable	30	70				
1378	29/07/2024	\$ 138.31		0.7	1.480	2.135	40.941	Stable	30	70				
1379	30/07/2024	\$ 135.67		2.64	1.480	1.921	43.513	Stable	30	70				
1380			=IF(B1379-B1378>0,(B1379-B1378),"											
1381			=IF(B1379-B1378<0,ABS(B1379-B1378),"											
1382	5		=IF(ISERROR(AVERAGE(C1365:C1379)),0,AVERAGE(C1365:C1379))											
1383			=IF(ISERROR(AVERAGE(D1365:D1379)),0,AVERAGE(D1365:D1379))											
1384			=100-(100/(1+(E1379/F1379)))											
1385			=IF(G1379<30,"The price will go UP",IF(G1379>70,"The price will go DOWN", "Stable"))											

Fig. A5.1

Cell C1379 has formula =IF(B1379-B1378>0,(B1379-B1378), ""). It just checks if the last price is higher than the previous one. If it is, it will show the difference between these two neighbouring prices. Wherever you see the value in this column C, this represents a daily gain in price.

Cell D1379 has formula =IF(B1379-B1378<0,ABS(B1379-B1378), ""). It just checks if the last price is lower than the previous one. If it is, it will show an absolute difference between these two neighbouring prices. Wherever you see the value in column D, this represents a daily loss in price.

The following two cells E1379 and F1379 have identical formulae. E1379 contains $=\text{AVERAGE}(J1365:J1379)$, but it has been modified to account for unexpected zeros, this is why this cell contains $=\text{IF}(\text{ISERROR}(\text{AVERAGE}(C1365:C1379)),0,\text{AVERAGE}(C1365:C1379))$. Same approach with F1379. These cells calculate the average gains and average absolute losses over the period of 14 days. If you go a row above to cells E1378 and F1378, you will see the same formulae, but for the corresponding 14 days. These are effectively 14-day moving averages for gains and losses.

Finally, cell G1379 contains an Excel version of the RSI equation. The formula is $=100-(100/(1+(E1379/F1379)))$. You see many different values in column G.

Cell H1379 is just using the words to describe the content of the neighbouring cell to the left, i.e. the value of the RSI index. The formula is $=\text{IF}(G1379<30,"The price will go UP",\text{IF}(G1379>70,"The price will go DOWN","Stable"))$. These are the rules we mentioned at the beginning of this Appendix as to how to interpret the RSI index.

In order to produce an RSI graph, we had to insert the values of 30 in column I and value of 70 in column J in Fig. A5.1. We can then create a line graph containing just RSI values from column G. After that, we can copy into the chart the values from columns I and J, which will give us the final chart, as in Fig. A5.2.




Fig. A5.2

If we copy values from column B (closing price values) into this chart and then change the secondary axis just to apply to these closing values, we get a chart as below in Fig. A5.3.

This chart enables a comparison between the actual movements of the closing price and the RSI.




Fig. A5.3

A6 How to calculate turning points and how to create the turning point chart

We'll break this down into several columns to make it easier to understand.

	A	B	C	D	E	F	G	H	I	J	
1	🏠 ORACLE CORPORATION (XNYS:ORCL)						L=	1.96			
2	Date	Close	UL	LL	IF	Clean Change	Turn				
1360	02/07/2024	\$ 143.28	144.06	136.69	Change UP						
1361	03/07/2024	\$ 144.38	145.33	137.07	Change UP						
1362	05/07/2024	\$ 144.83	145.75	139.10	Change UP						
1363	08/07/2024	\$ 145.03	146.12	140.59	Change UP						
1364	09/07/2024	\$ 140.68	145.87	142.38	Change DOWN	Change DOWN	\$ 140.68				
1365	10/07/2024	\$ 142.07	147.14	140.14	Change DOWN						
1366	11/07/2024	\$ 142.76	147.17	139.62	Change DOWN						
1367	12/07/2024	\$ 144.77	146.71	139.44	Change DOWN						
1368	15/07/2024	\$ 143.07	146.67	139.46	Change DOWN						
1369	16/07/2024	\$ 142.61	145.59	139.75	Change DOWN						
1370	17/07/2024	\$ 139.26	145.06	141.05	Change DOWN						
1371	18/07/2024	\$ 138.03	146.42	138.57	Change DOWN						
1372	19/07/2024	\$ 138.56	147.04	136.05	Change DOWN						
1373	22/07/2024	\$ 140.17	144.93	135.68	Change DOWN						
1374	23/07/2024	\$ 143.11	143.26	136.20	Change DOWN						
1375	24/07/2024	\$ 138.77	143.75	135.90	Change DOWN						
1376	25/07/2024	\$ 137.82	143.74	135.71	Change DOWN						
1377	26/07/2024	\$ 139.01	143.79	135.58	Change DOWN						
1378	29/07/2024	\$ 138.31	143.78	135.77	Change DOWN						
1379	30/07/2024	\$ 135.67	143.56	135.25	Change DOWN						
1380			=AVERAGE(B1374:B1378)+\$G\$1*STDEV.S(B1374:B1378)								
1381			=AVERAGE(B1374:B1378)-\$G\$1*STDEV.S(B1374:B1378)								
1382	5		=IF(B1379>C1379,"Change UP",IF(B1379<D1379,"Change DOWN",E1378))								
1383			=IF(E1379=E1378,"",E1379)								
1384			=IF(F1379<>"",B1379,"")								

Fig. A6.1

In Fig. A6.1, we created new columns C:G, and a new cell G1, where G1 is the value of standard deviation, and we are using here value of 1.96, which is equivalent to the z-score.

The formula we want to reproduce is:

$$\bar{y}_D \pm L \sigma_D$$

Where,

\bar{y}_D = The mean value of the last 5 closing prices (D=5)

σ_D = Standard deviation of the last 5 closing prices (D=5)

L = Number of standard deviations expressed as standardized z-score

We are effectively creating an interval between the Upper and Lower Limit values. As this value is expressed as a 5-day moving average, it changes continuously. Upper Limit values are given in column C and Lower Limit values in column D.

Cell C1379 has a formula: =AVERAGE(B1374:B1378)+ \$G\$1*STDEV.S(B1374:B1378). This is effectively the upper confidence level for the previous five observations. We take the five-point average and add it to the standard deviation for the same interval, multiplied by the value from G1, which determines how wide the interval will be. Cell D1379 does the same, except that we subtract the expression from the average.

We have effectively created a confidence interval based on the previous five closing price values. The sensitivity, or the width of this interval is determined by the value of standard deviation and the value from G1, which is the z-value. If you insert 1 in this cell, the interval will be narrower and more sensitive to changes. We opted for 1.96, which corresponds with a 95% confidence interval.

Column E1379 in Fig. A6.1 contains formula =IF(B1379>C1379,"Change UP",IF(B1379<D1379,"Change DOWN",E1378)). This is nothing but a bit longer Excel expression saying: "If the current price in cell B1379 is above the interval defined by C1379, insert the phrase "Change UP". If the value is below the interval in D1379, insert the phrase "Change DOWN"".

If you look up this column E, you will see a series of interchanging phrases. They indicate how many times the current value was climbing and how many times it was going down. This is good information, but we want to indicate just the point in time when the change took place, i.e. the turning point.

We solved this problem in column F in Fig. A6.1, where in cell F1379 we have a formula =IF(E1379=E1378,"",E1379). It effectively says to ignore the repeated labels and insert the label only when the turning point occurs.

Now we have clear turning points labelled, we actually need a value of the closing stock at these points. This is achieved in column G in Fig. A6.1 with a formula =IF(F1379<>"",B1379,""). In other words, where you see the label in column F, return the corresponding value from column B, i.e. the closing price. We now have all the elements to create the turning point graph.

To create the graph, we highlight the last 90 days in columns A and B in Fig. A6.1, which are the dates and closing values, and select a line graph. Initially, the graph looks like the one in Fig. A6.2.




Fig. A6.2

For the same 90-day period we now select the values from column G, which is G1293:G1379 and copy/ paste these values into the graph we just created. The new graph will look like the one in Fig. A6.3.




Fig. A6.3

To clean up the graph, right-click anywhere on the graph and when the dialogue box opens, select Change Chart Type... (see Fig. A6.4).




Fig. A6.4

Now select the Combo graph and ensure that the closing prices (Series 1) are a line graph and the turning point values (Series 2) are clustered columns. Click OK. See Fig. A6.4.




Fig. A6.4

The final result is a graph as shown in Fig. A6.5.




Fig. A6.5

To make it a bit more visible, change the range of the y-axis to 110 as a minimum. Depending on the values of the stocks you are handling, you might need to use a different minimum value. The chart now looks as in Fig. A6.6.




Fig. A6.6

Every vertical line indicates when the turning point has begun. It is delayed a bit behind the actual turning point, and the reason for this is that we are using the previous five observation statistics (the mean and the standard deviation) to judge if the turning point started. This inevitably generates a small delay between 1 and 3 days.

As it happens, from 09/07/2024 we have indications that the prices started dropping. If we are determined to buy this stock, we should wait until the next turning point indicates that the prices started going up. Whenever this happens, this is the point in time to buy the shares.

There is an alternative way to present turning points. For this, we could use a line graph as in Fig. A7.7.




Fig. A7.7

The blue line in Fig. A7.7 represents the closing prices over the last 90 days. The two dotted lines are the Upper and Lower Limits. The Upper Limit is usually above the solid blue price line and the Lower Limit is usually below the solid blue price line.

The turning point means the change, in other words, the turn downwards is always followed by the turn upwards, etc. We used red circles to indicate a downward turn and green circles to indicate an upward turn.

When the blue line crosses the lower dotted line for the first time in an interval, this represents a downward turning point. When the blue line crosses the upper dotted line for the first time in an interval after a downward turning point, this represents an upward turning point. As you can see from the graph, they succeed each other.

Our graph does not show that the closing price is getting close to the upper dotted line, which means that the upward trend is not imminent. In this case, we said that we should hold back and not buy the shares yet. They still might continue to fall.

A7 P.S. What actually happened?

I completed the first pass of this material on 30/07/2024 and this represents the “current day”. However, it took me another several weeks to fix all the errors and edit the material. This also allowed me to see if my recommendations were sound. Here is what happened.

The closing price on 30/07/2024 was \$135.67 and the analysis indicated that we should go and buy this stock, but the question was when?

Looking at all our charts, due to some oscillations in stock price, we could gain a lot if we waited for the right moment to buy. On 30/07/2024 when the analysis was closed, I said that we should

wait a few days to see the end of the downward trend. When we hit the next turning point and the prices start going up, this would be the time to buy.

When I refreshed the data at the end of 22/08/2024, the price was \$138.08. This indicates a small increase when compared to \$135.67 on 30/07/2024. However, between 30/07/2024 and 22/08/2024 the price dropped to a low point of \$125.09 and then gradually recovered to the current level of \$138.08.

Let's look at the updated RSI chart first in Fig.A7.1. RSI indices beyond the previous "current date", i.e. 30/07/2024 are coloured in purple.




Fig. A7.1

According to RSI in Fig. A7.2, the stocks showed a trend towards the overvalue point until 01/08/2024 and then started to show a potential move towards the undervalue point. In fact, on 07/08/2024 the closing prices hit rock bottom for this downward interval and recorded the closing price of \$125.90. However, the RSI index did not indicate the undervalue level until 21/08/2024.

The lesson here is that RSI will indicate a change in direction, but the actual change always happens a few days after that. In other words, look for the changes in RSI, but do not pull the trigger immediately and buy or sell. Wait a few days and then make a judgment call.

The next indicator, which is the turning point graph, might help with pinning down more precisely on what day after the change in RSI we should actually buy or sell the shares. Fig. A7.2 shows the updated turning point chart. Again, the closing prices after our "current day" on 30/07/2024 are given in purple.




Fig. A7.2

On 09/07/2024, we had a signal that the stocks had started going down. It was only on 21/08/2024 that we got a signal that the stocks are potentially again changing direction. This was an interval during which we should have bought our shares. Unfortunately, nothing indicated that we should have placed an order on 07/08/2024 when the price was the lowest during this period. Effectively, we could have missed the boat to buy the shares at their lowest level during this period. What went wrong?

Well, nothing! We just should have modified one parameter, which is the cell G1 where we had 1.96 for the z-value. This value is OK if we are looking at historical data and we do not want to see too many turning points. However, when we are in a tactical mode and looking to buy or sell, we need to make the Upper and Lower boundary narrower, i.e. make this method more sensitive.

If we change cell G1 to just 1, we get a chart as in Fig. A7.3.




Fig. A7.3

We can now see that the turning point signal happened on 09/08/2024. On that day the price was \$132.06. We would have still missed the lowest price of \$125.90, but at least we are as close as we can get to this price.

So, what is the message here?

We said that buying and selling shares is equal part art as it is a science. There is no perfect method to predict what to do on any given day. However, with the methods we described here, this gets you as close as you can get to the perfect point in time when to buy or sell.

The closing note is that buying and selling shares is a form of hunting, but a specific form of hunting that used to be called trapping. You set up an ambush, i.e. define the circumstances under which you will execute the purchase or the sell, and then just wait until these circumstances are met and you are comfortable that you have the catch in the right condition, i.e. at the right price. In other words, it is a game that requires patience and some skills.